Prof. Dr. Jürgen Giesl

Carsten Fuhs, Carsten Otto, Thomas Ströder

Bachelor/Master Exam Version V3B

First Name: _____

Last Name: _____

Immatriculation Number: _____

Course of Studies (please mark exactly one):

- ○ **Informatik Bachelor**
- ○ **Mathematik Bachelor (for Master)**
- ○ **Mathematik Master**
- ○ **Other:** _____

	Maximal Points	Achieved Points
Exercise 1	10	
Exercise 2	16	
Exercise 3	10	
Exercise 4	5	
Exercise 5	10	
Exercise 6	9	
Total	60	
Grade	-	

Instructions:

- On every sheet please give your **first name**, **last name**, and **immatriculation number**.
- You must solve the exam **without** consulting any **extra documents** (e.g., course notes).
- Make sure your answers are readable. Do not use **red or green pens or pencils**.
- Please answer the exercises on the **exercise sheets**. If needed, also use the back sides of the exercise sheets.
- Answers on extra sheets can only be accepted if they are clearly marked with your name, your immatriculation number, and the **exercise number**.
- **Cross out** text that should not be considered in the evaluation.
- Students that try to cheat **do not pass** the exam.
- At the end of the exam, please return **all sheets together with the exercise sheets**.

Exercise 1 (Theoretical Foundations): (6 + 4 = 10 points)

Let

$$\varphi \;=\; \mathtt{p}(0) \;\wedge\; \exists X\, \neg \mathtt{p}(X) \;\wedge\; \forall Y\, (\,\mathtt{p}(Y) \rightarrow \mathtt{p}(\mathtt{s}(\mathtt{s}(Y)))\,) \quad \text{and}$$

$$\psi \;=\; \exists Z\, \forall U\, (\,\mathtt{p}(U) \rightarrow \mathtt{p}(\mathtt{s}(Z))\,)$$

be formulas over the signature (Σ, Δ) with $\Sigma = \Sigma_0 \cup \Sigma_1$, $\Sigma_0 = \{\mathtt{0}\}$, $\Sigma_1 = \{\mathtt{s}\}$, and $\Delta = \Delta_1 = \{\mathtt{p}\}$.

a) Prove that $\varphi \models \psi$ by means of resolution.

Hint: First transform the formula $\varphi \wedge \neg\psi$ into a satisfiability-equivalent Skolem normal form, and then convert this Skolem normal form to an equivalent clause set.

b) Explicitly give a Herbrand model of the formula φ (i.e., specify a carrier and a meaning for all function and predicate symbols). You do not have to provide a proof for your answer.

Solution:

a)

$$\begin{aligned}
& \varphi \wedge \neg\psi \\
\Leftrightarrow\;& \mathtt{p}(0) \;\wedge\; \exists X\, \neg \mathtt{p}(X) \;\wedge\; \forall Y\, (\,\mathtt{p}(Y) \rightarrow \mathtt{p}(\mathtt{s}(\mathtt{s}(Y)))\,) \;\wedge\; \neg\exists Z\, \forall U\, (\,\mathtt{p}(U) \rightarrow \mathtt{p}(\mathtt{s}(Z))\,) \\
\Leftrightarrow\;& \mathtt{p}(0) \;\wedge\; \exists X\, \neg \mathtt{p}(X) \;\wedge\; \forall Y\, (\,\neg\mathtt{p}(Y) \vee \mathtt{p}(\mathtt{s}(\mathtt{s}(Y)))\,) \;\wedge\; \forall Z\, \exists U\, \neg(\,\neg\mathtt{p}(U) \vee \mathtt{p}(\mathtt{s}(Z))\,) \\
\Leftrightarrow\;& \mathtt{p}(0) \;\wedge\; \exists X\, \neg \mathtt{p}(X) \;\wedge\; \forall Y\, (\,\neg\mathtt{p}(Y) \vee \mathtt{p}(\mathtt{s}(\mathtt{s}(Y)))\,) \;\wedge\; \forall Z\, \exists U\, (\,\mathtt{p}(U) \wedge \neg\mathtt{p}(\mathtt{s}(Z))\,) \\
\Leftrightarrow\;& \exists X\, \forall Y\, \forall Z\, \exists U\, (\,\mathtt{p}(0) \;\wedge\; \neg\mathtt{p}(X) \;\wedge\; (\,\neg\mathtt{p}(Y) \vee \mathtt{p}(\mathtt{s}(\mathtt{s}(Y)))\,) \;\wedge\; \mathtt{p}(U) \wedge \neg\mathtt{p}(\mathtt{s}(Z))\,)
\end{aligned}$$

We obtain the following formula θ in Skolem normal form, which is satisfiability-equivalent to $\varphi \wedge \neg\psi$ (by dropping existential quantification, replacing X by $\mathtt{c}$ and replacing U by $\mathtt{f}(Y, Z)$).

$$\theta = \forall Y\, \forall Z\, (\,\mathtt{p}(0) \;\wedge\; \neg\mathtt{p}(\mathtt{c}) \;\wedge\; (\,\neg\mathtt{p}(Y) \vee \mathtt{p}(\mathtt{s}(\mathtt{s}(Y)))\,) \;\wedge\; \mathtt{p}(\mathtt{f}(Y, Z)) \wedge \neg\mathtt{p}(\mathtt{s}(Z))\,)$$

An equivalent clause set for the Skolem normal form θ of $\varphi \wedge \neg\psi$ is:

$$\{\quad \{\mathtt{p}(0)\}, \{\neg\mathtt{p}(\mathtt{c})\}, \{\neg\mathtt{p}(Y), \mathtt{p}(\mathtt{s}(\mathtt{s}(Y)))\}, \{\mathtt{p}(\mathtt{f}(Y, Z))\}, \{\neg\mathtt{p}(\mathtt{s}(Z))\} \quad \}$$

We perform resolution on this clause set to show unsatisfiability of θ and hence also of $\varphi \wedge \neg\psi$.



Hence, we have proven $\varphi \models \psi$. □

b) We have $S \models \varphi$ for the Herbrand structure $S = (\mathcal{T}(\Sigma), \alpha)$ with $\alpha_0 = 0$, $\alpha_{\mathtt{s}}(t) = \mathtt{s}(t)$, and $\alpha_{\mathtt{p}} = \{\mathtt{s}^{2i}(0) \mid i \in \mathbb{N}\}$.

Exercise 2 (Procedural Semantics, SLD tree): (7 + 9 = 16 points)

Consider the following Prolog program $\mathcal{P}$ which can be used to divide two numbers represented by terms built from $0 \in \Sigma_0$ and $\mathtt{s} \in \Sigma_1$ (i.e., $0 \mathrel{\hat{=}} 0, 1 \mathrel{\hat{=}} \mathtt{s}(0), 2 \mathrel{\hat{=}} \mathtt{s}(\mathtt{s}(0)), \ldots$):

```
div(X, 0, Z) :- !, fail.
div(0, s(Y), Z) :- eq(Z, 0), !.
div(X, Y, s(Z)) :- sub(X, Y, U), div(U, Y, Z).
sub(0, Y, 0).
sub(s(X), 0, s(X)).
sub(s(X), s(Y), Z) :- sub(X, Y, Z).
eq(X, X).
```

a) The program $\mathcal{P}'$ results from $\mathcal{P}$ by **removing the cuts**. Consider the following query:

```
?- div(s(0), s(0), s(s(0))).
```

For the logic program $\mathcal{P}'$, i.e., **without the cuts**, please show a successful computation for the query above (i.e., a computation of the form $(G, \varnothing) \vdash^{+}_{\mathcal{P}'} (\square, \sigma)$ where $G = \{\neg \mathtt{div}(\mathtt{s}(0), \mathtt{s}(0), \mathtt{s}(\mathtt{s}(0)))\}$). You do not need to extend the substitution, i.e., it suffices to write $\varnothing$ in each step.

b) Please give a graphical representation of the SLD tree for the query `?- div(s(s(0)), s(0), X).` in the program $\mathcal{P}$ (i.e., **with the cuts**). Also give all answer substitutions explicitly.

We repeat the program below.

```
div(X, 0, Z) :- !, fail.
div(0, s(Y), Z) :- eq(Z, 0), !.
div(X, Y, s(Z)) :- sub(X, Y, U), div(U, Y, Z).
sub(0, Y, 0).
sub(s(X), 0, s(X)).
sub(s(X), s(Y), Z) :- sub(X, Y, Z).
eq(X, X).
```

Solution:

a)

$$
\begin{aligned}
& (\{\neg \mathtt{div}(s(0), s(0), s(s(0)))\}, \varnothing) \\
\vdash_{\mathcal{P}'} & (\{\neg \mathtt{sub}(s(0), s(0), U), \neg \mathtt{div}(U, s(0), s(0))\}, \varnothing) \\
\vdash_{\mathcal{P}'} & (\{\neg \mathtt{sub}(0, 0, U), \neg \mathtt{div}(U, s(0), s(0))\}, \varnothing) \\
\vdash_{\mathcal{P}'} & (\{\neg \mathtt{div}(0, s(0), s(0))\}, \varnothing) \\
\vdash_{\mathcal{P}'} & (\{\neg \mathtt{sub}(0, s(0), U'), \neg \mathtt{div}(U', s(0), 0)\}, \varnothing) \\
\vdash_{\mathcal{P}'} & (\{\neg \mathtt{div}(0, s(0), 0)\}, \varnothing) \\
\vdash_{\mathcal{P}'} & (\{\neg \mathtt{eq}(0, 0)\}, \varnothing) \\
\vdash_{\mathcal{P}'} & (\square, \varnothing)
\end{aligned}
$$

b)

In this representation, the nodes and edges deleted by the cut are shown with a gray background and dashed edges, respectively.



Exercise 3 (Definite Logic Programming): (10 points)

A directed graph is a pair (V, E) where V is the set of vertices and $E \subseteq V \times V$ is the set of edges. We represent such a graph by two lists where the first list contains the names of the nodes and the second list contains an element `e(`n_1`,`n_2`)` iff there is an edge from n_1 to n_2 in the represented graph. Both lists do not contain duplicates. For example, consider the following graph $(\{1, 2, 3\}, \{(1, 2), (3, 1), (2, 3)\})$:



This graph is represented by the two lists `[1,2,3]` and `[e(1,2),e(3,1),e(2,3)]`.

Implement the predicate `hamiltonian/2` in Prolog. A call `hamiltonian(`t_1`,`t_2`)` works as follows. If both t_1 and t_2 are finite lists representing a directed graph as described above, the call succeeds iff there is a hamiltonian cycle in the represented graph. If t_1 and t_2 are not of the form described above, then `hamiltonian(`t_1`,`t_2`)` may behave arbitrarily. You **must not use** any built-in predicates in this exercise. Note that meta-programming (e.g., using variables without a surrounding predicate) also uses built-in predicates (implicitly) and is, thus, not allowed in this exercise.

A hamiltonian cycle exists in a directed graph iff there is a path through the graph starting and ending in the same node (thus, visiting this node exactly twice) and visiting all other nodes exactly once. The empty graph has no hamiltonian cycle. The following example calls to `hamiltonian/2` illustrate its definition:

- `?- hamiltonian([1,2,3],[e(1,2),e(3,1),e(2,3)]).` succeeds with the empty answer substitution
- `?- hamiltonian([1,2,3],[e(1,2),e(2,1),e(1,3),e(3,1)]).` fails

Hint: Use a helper predicate `tour/4` *with the following four arguments: (1) the node where you currently are, (2) the list of nodes yet to visit, (3) the node you have to reach in the end, and (4) the list of edges in the graph. Moreover, it can be helpful to define a predicate* `choice/3` *for the non-deterministic choice of the next node to visit. Here,* `choice(`t_1`,`t_2`,`t_3`)` *is true if t_1 is a list containing t_2 and t_3 is the list t_1 where the element t_2 was deleted.*

Solution:

```
hamiltonian([N|NS],E) :- tour(N,NS,N,E).

tour(L,[],T,E) :- edge(L,T,E).
tour(L,[X|XS],T,E) :- choice([X|XS],Y,YS),
                      edge(L,Y,E),
                      tour(Y,YS,T,E).

choice([X|XS],X,XS).
choice([X|XS],Y,[X|YS]) :- choice(XS,Y,YS).

edge(L,T,[e(L,T)|_]).
edge(L,T,[_|E]) :- edge(L,T,E).
```

.

Exercise 4 (Arithmetic): (5 points)

Implement the predicate `mean/2` in Prolog. A call of `mean(`t_1`,`t_2`)` works as follows. If t_1 is a finite non-empty list of integers, then t_2 is unified with the *rounded mean value* of all numbers in t_1. If t_1 is the empty list, `mean/2` fails. If t_1 is not a finite list of integers, `mean/2` may behave arbitrarily.

We define the *rounded mean value* of $n > 0$ integers $a_1, \ldots, a_n$ as $\left\lfloor \frac{\sum_{i=1}^{n} a_i}{n} \right\rfloor$ if $\frac{\sum_{i=1}^{n} a_i}{n}$ is positive and $\left\lceil \frac{\sum_{i=1}^{n} a_i}{n} \right\rceil$ if $\frac{\sum_{i=1}^{n} a_i}{n}$ is negative.

The following example calls to `mean/2` illustrate its definition:

- `?- mean([],X).` fails
- `?- mean([1,2,3],1).` fails
- `?- mean([1,2,3],X).` succeeds with the answer substitution `X = 2`
- `?- mean([1,2],1).` succeeds with the empty answer substitution
- `?- mean([1,-2],X).` succeeds with the answer substitution `X = 0`

Hint: In Prolog, the term `X // Y` *can be used to compute* $\left\lfloor \frac{X}{Y} \right\rfloor$ *if* $\frac{X}{Y}$ *is positive and* $\left\lceil \frac{X}{Y} \right\rceil$ *if* $\frac{X}{Y}$ *is negative.*

Solution:

```
mean(List,Mean) :- sumMean(0,0,List,Mean).

sumMean(Length,Sum,[],Mean) :- Length > 0,
                               Mean is Sum // Length.
sumMean(Length,Sum,[X|XS],Mean) :- NextLength is Length + 1,
                                   NextSum is Sum + X,
                                   sumMean(NextLength,NextSum,XS,Mean).
```

Exercise 5 (Meta-Programming): (10 points)

Implement the predicate `fold/4` in Prolog. A call of `fold(`t_1`,`t_2`,`t_3`,`t_4`)` works as follows. If t_1 is a constant $f \in \Sigma_0$ and t_3 has the form $[a_1, \ldots, a_n]$ with $n > 0$ (i.e. t_3 is a finite non-empty list), then the calls $f(t_2, a_1, X_1), f(X_1, a_2, X_2), \ldots, f(X_{n-2}, a_{n-1}, X_{n-1}), f(X_{n-1}, a_n, t_4)$ with fresh Prolog variables $X_1, \ldots, X_{n-1}$ are executed. That means we assume that there is also a predicate symbol $f \in \Delta_3$ (with the same name as $f \in \Sigma_0$). Thus, `fold(`f`,`t_2`,`$[a_1, \ldots, a_n]$`,`t_4`)` succeeds iff the query $f(t_2, a_1, X_1), f(X_1, a_2, X_2) \ldots, f(X_{n-2}, a_{n-1}, X_{n-1}), f(X_{n-1}, a_n, t_4)$ succeeds. If t_1 or t_3 are not of the form described above, `fold/4` may behave arbitrarily.
For example, the query `?- fold(foo,d,[a,b,c],X).` is evaluated by executing the three calls `foo(d,a,X1)`, `foo(X1,b,X2)` and `foo(X2,c,X)`. The query `?- fold(foo,b,[a],r).` is evaluated by the call `foo(b,a,r)`.

Hint: You may use the built-in predicate `=../2`.

Solution:

```
fold(F,A,[E],Z) :- !,
                   C =.. [F,A,E,Z],
                   call(C).
fold(F,A,[E|ES],Z) :- C =.. [F,A,E,X],
                      call(C),
                      fold(F,X,ES,Z).
```

Exercise 6 (Difference Lists): (9 points)

Consider the following logic program $\mathcal{P}$.

```
q(X) :- p(X - []).

p([X|Y] - Y).
p([X|Y] - Z) :- p(Y - [X|Z]).
```

Explicitly give the set of all ground terms t for which the query `?- q(t).` succeeds. State a formal definition of the set in closed form (i.e., do not use a recursive definition). You do not have to provide a proof for your answer.

Solution:

$\{[t_1, \ldots, t_{n-1}, t_n, t_{n-1}, \ldots, t_1] \mid n > 0, t_i \text{ is a ground term for all } i \in \{1, \ldots, n\}\}$